Math 231 tutorial

Solve Ax=b using Gaussian elimination

```
A = matrix([[1,2,3],[4,5,6],[7,8,2]])
b = vector([-1,4,-7])
Aaug = A.augment(b); view(Aaug.rref())
```

$$\begin{pmatrix} 1 & 0 & 0 & \frac{139}{21} \\ 0 & 1 & 0 & -\frac{152}{21} \\ 0 & 0 & 1 & \frac{16}{7} \end{pmatrix}$$

Compute the matrix inverse

```
Ainv = A.inverse; show(Ainv())
```

$$\begin{pmatrix} -\frac{38}{21} & \frac{20}{21} & -\frac{1}{7} \\ \frac{34}{21} & -\frac{19}{21} & \frac{2}{7} \\ -\frac{1}{7} & \frac{2}{7} & -\frac{1}{7} \end{pmatrix}$$

```
show(A*Ainv())
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Compute the determinant

```
show(A.determinant())
```

$$21$$

Compute the matrix inverse using Gaussian elimination

```
II = identity_matrix(3)
AaugII = A.augment(II); show(AaugII.rref())
```

$$\begin{pmatrix} 1 & 0 & 0 & -\frac{38}{21} & \frac{20}{21} & -\frac{1}{7} \\ 0 & 1 & 0 & \frac{34}{21} & -\frac{19}{21} & \frac{2}{7} \\ 0 & 0 & 1 & -\frac{1}{7} & \frac{2}{7} & -\frac{1}{7} \end{pmatrix}$$

Compute the eigenvalues and eigenvectors using exact arithmetic (if possible)

```
A.eigenvectors_right()
```

```
[(-4.442318317846453?, [(1, 1.265019102818151?,
-2.657452174494252?)], 1),
 (-0.3689912718946151?, [(1, -0.921859388091059?,
0.1582425014291674?)], 1),
 (12.811309589741069?, [(1, 2.338658467091090?,
2.377997551852963?)], 1)]
```

Compute the eigenvalues and eigenvectors numerically

```
Ardf = matrix(RDF,[[1,2,3],[4,5,6],[7,8,2]])
Ardf.eigenvectors_right()
```

```
[(12.811309589741061,
  [(-0.28719278795632724, -0.6716458452415609,
-0.6829437466699737)],
  1),
 (-0.3689912718946157,
  [(-0.7303229535101788, 0.673255071031748, -0.11556813101458799)],
  1),
 (-4.4423183178464525,
  [(-0.32170600725573373, -0.40696424466985737,
0.8549183285296124)],
  1)]
```

Eigenvalues and eigenvectors using exact arithmetic (if possible)

```
Mar = matrix([[2/10,0,1/100],[8/10,5/10,0],[0,5/10,99/100]])
Mar.eigenvectors_right()
```

```
[(1, [
  (1, 8/5, 80)
  ], 1),
 (0.2184101109882784?, [(1, -2.841011098827837?,
1.841011098827838?)], 1),
 (0.4715898890117217?, [(1, -28.15898890117217?,
27.15898890117217?)], 1)]
```

Rescale the eigenvector so that the entries sum to one

```
show(vector([1, 8/5, 80])/(1+8/5+80))
```

$$\left(\frac{5}{413}, \frac{8}{413}, \frac{400}{413}\right)$$

```
B = matrix([[2*I,3],[-2+5*I,-3*I]])
Binv = B.inverse; show(Binv())
```

$$\begin{pmatrix} -\frac{4}{41}i + \frac{5}{41} & -\frac{5}{41}i - \frac{4}{41} \\ -\frac{10}{123}i + \frac{11}{41} & \frac{8}{123}i - \frac{10}{123} \end{pmatrix}$$

Solve Ax=b where the coefficients are variables. Simplify each component of the solution vector.

```
var("a,b,c,V")
C = matrix([[-2*a,a,-b,0],[a,-2*a,0,-b],[b,0,-2*a,a],[0,b,a,-2*a]])
cf = vector([a*V*c,0,0,0])
cans = C\cf; cans
```

```
(-2/3*V*c + 2/3*V*b^2*c/((3*a + b^2/a)*a) - 2/3*(V*b*c + V*(3*a -
b^2/a)*b*c/(3*a + b^2/a))*((3*a - b^2/a)*b/((3*a + b^2/a)*a) +
b/a)/((3*a - b^2/a)^2/(3*a + b^2/a) - 12*a - 4*b^2/a), -1/3*V*c +
1/3*V*b^2*c/((3*a + b^2/a)*a) - 1/3*(V*b*c + V*(3*a -
b^2/a)*b*c/(3*a + b^2/a))*((3*a - b^2/a)*b/((3*a + b^2/a)*a) +
4*b/a)/((3*a - b^2/a)^2/(3*a + b^2/a) - 12*a - 4*b^2/a), -V*b*c/(3*a
+ b^2/a) + (V*b*c + V*(3*a - b^2/a)*b*c/(3*a + b^2/a))*(3*a -
b^2/a)/(((3*a - b^2/a)^2/(3*a + b^2/a) - 12*a - 4*b^2/a)*(3*a +
b^2/a)), 2*(V*b*c + V*(3*a - b^2/a)*b*c/(3*a + b^2/a))/((3*a -
b^2/a)^2/(3*a + b^2/a) - 12*a - 4*b^2/a))
```

```
show(factor(cans[0]))
```

$$-\frac{2\left(3\,a^2 + b^2\right)Va^2c}{(9\,a^2 + b^2)(a^2 + b^2)}$$

```
show(factor(cans[1]))
```

$$-\frac{\left(3\,a^2 - b^2\right)Va^2c}{(9\,a^2 + b^2)(a^2 + b^2)}$$

```
show(factor(cans[2]))
```

$$-\frac{\left(5\,a^2 + b^2\right)Vabc}{(9\,a^2 + b^2)(a^2 + b^2)}$$

```
show(factor(cans[3]))
```

$$-\frac{4\,Va^3bc}{(9\,a^2+b^2)(a^2+b^2)}$$

```
show(factor(C.determinant()))
```

$$\left(9\,a^2+b^2\right)\left(a^2+b^2\right)$$

Variation of parameters solution for first-order scalar ODE

```
var("a,w,t")
f(t) = exp(a*t)*sin(w*t)
F(t) = f.integrate(t)
yp(t) = exp(-a*t)*F(t); show(yp(t))
```

$$-\frac{w\cos\left(tw\right)-a\sin\left(tw\right)}{a^2+w^2}$$

Partial fraction expansion

```
h = x^3/((x+1)*(x+3)*(x^2+9)); show(h.partial_fraction())
```

$$\frac{3\left(x-6\right)}{10\left(x^2+9\right)}+\frac{3}{4\left(x+3\right)}-\frac{1}{20\left(x+1\right)}$$

Laplace transform and inverse Laplace transform

```
var("s,t")
f = t^2 + exp(-2*t) - sin(3*t); show(f.laplace(t,s))
```

$$-\frac{3}{s^2+9}+\frac{1}{s+2}+\frac{2}{s^3}$$

```
g = (s^3-3*s^2+5*s+1)/((s+2)*(s+5)*(s^2+6*s+13))
show(inverse_laplace(g,s,t))
```

$$-\frac{1}{5}\left(32\,\cos\left(2\,t\right)-19\,\sin\left(2\,t\right)\right)e^{\left(-3\,t\right)}-\frac{29}{15}\,e^{\left(-2\,t\right)}+\frac{28}{3}\,e^{\left(-5\,t\right)}$$